

طرق الإهلاك بالاكسيل

Depreciation in Excel



بِسْمِ اللَّهِ الرَّحْمَنِ الرَّحِيمِ

فَأَمَّا الزُّبَدُ فَيَذْهَبُ جُفَاءً وَأَمَّا

مَا يَنْفَعُ النَّاسَ فَيَمْكُثُ فِي

(الأَرْضِ)

صَدَقَ اللَّهُ الْعَظِيمُ

تعريف بالكاتب



محمد الريفى

Mohamed Elrifly

Email: xls.financial@gmail.com

Tel: [0201026059594](tel:0201026059594)

- خبير ومدرّب EXCEL مالي ومحاسبي
- حاصل على بكالوريوس التجارة ودبلوم الدراسات العليا في المحاسبة والمراجعة جامعة القاهرة
- خبير اكسيل معترف به في مجتمع مايكروسوفت اكسيل وحصلت على العديد من الجوائز منها (MCC(Content Creator)
- حاصل على شهادة خبير اكسيل MOS expert excel 2010
- خبير اكسيل معتمد في العديد من المنتديات العربية والأجنبي
- مؤسس مدونة خبير اكسيل وصفحة وجروب خبير اكسيل على الفيس بوك
- حاصل على (IC3(Internet & Computing Core Certification)
- قدمت مواضيع للماليين مثل تحليل الحساسية بالاكسيل وتحليل التنبؤات بالاكسيل وتحليل التعادل.... الخ وموضوعات اخرى تهم المحاسبين والمدراء الماليين .

وما توفيقى الا بالله

مقدمة

بسم الله الرحمن الرحيم

الحمد لله رب العالمين والصلاة والسلام على اشرف الأنبياء والمرسلين النبي الكريم سيدنا محمد وعلى آله واصحابه الطيبين الطاهرين اجمعين .

تعتبر عملية الاهلاك توزيعا لتكلفة الأصل على مدار عمره الإنتاجي ولقد وفرلنا برنامج الاكسيل دوال مختلفة وتقوم بعمل حسابات متنوعة للاهلاك وحاولنا ابراز هذه الدوال ومكونات كل دالة وطريقة استخدامها كما وردت في دعم مايكروسوفت . وعرضنا الطرق اليدوية التي تحسب نتائج هذه الدوال ، ووجدنا ان كل دوال الاكسيل تحسب فترات كاملة او سنة كاملة لقسط الاهلاك ولكن لا تقوم بحساب الاهلاك لفترات جزئية وهذا الامر هو ما يحدث في الواقع العملي كأن يتم حساب اهلاك ابتداء من منتصف او أي فتره في السنة . وهي صيغ طويلة وأتمنى ان تقوم مايكروسوفت بعمل تحسين على هذه الدوال في المستقبل القريب .

وقد تناول هذه الموضوع العديد من الأساتذة الكرام وهذا الكتيب اقدمه للجميع لينضم الى ما قدمه السادة الافاضل جزاهم الله خيرا .

اسأل المولى عزوجل ان يتقبل من هذا العمل وان يجعله في ميزان الحسنات لى ولكل من يتعلم منه او يعلمه لغيرى

خبير اكسيل

محمد الريفى

في ابريل ٢٠١٧

المحتويات

المحتويات	من	الى
التعريف بالكاتب	3	3
مصطلحات الاهلاك	5	5
دوال الاهلاك	6	6
دالة القسط الثابت SLN	7	10
دالة القسط المتناقص المضاعف DDB	11	17
دالة مجموع ارقام السنوات SYD	18	23
طريقة وحدات الإنتاج	24	24
دالة القسط المتناقص الثابت DB	25	27
دالة القسط المتناقص المتغير VDB	28	31
مقارنة بين دوال الاهلاك	32	32
دالة AMORLINC النظام الفرنسي	33	34
دالة AMOREGRC النظام الفرنسي	34	35
امثلة عملية من موسوعة معايير المحاسبة	36	39
المراجع	40	40

بسم الله الرحمن الرحيم

حساب الاهلاك باستخدام الاكسيل

تعريف الاهلاك.

لا تكاد تخلو أي ميزانية من بند الاصول طويلة الاجل او الاصول الثابتة او الاصول غير المتداولة والتي تستخدم في تسهيل اعمال الشركة وتقدم منافع اقتصادية ومستقبلية لفترات زمنية اكثر من عام .لذا فلا بد من توزيع تكلفة هذه الاصول على العمر الإنتاجي لها او المتوقع الاستفادة منه وهو ما يطلق عليه الاستهلاك .

اذن فان الاهلاك Depreciation حسب معيار المحاسبة الدولي رقم ١٦

هو التوزيع المنتظم للقيمة القابلة للاستهلاك للأصل على مدار عمرة الإنتاجي المتوقع

والقيمة القابلة للاستهلاك:

هي عبارة عن تكلفة الاصل (-) مطروحا منها القيمة المتبقية (الخردة) في نهاية عمرة الإنتاجي . بفرض ان اذا كانت تكلفة اصل ما ب ١٠٠٠٠ ج وقيمة الخردة او النفاية ١٠٠٠ ج فان القيمة القابلة للاستهلاك هي ٩٠٠٠ ج

وتكلفة الاصل (cost):

بموجب المعيار فانه يتم تسجيل قيمة الاصل الثابت بالتكلفة وتتضمن التكلفة جميع التكاليف الضرورية التي يتحملها الاصل حتى يصبح جاهزا للاستخدام او للغرض الذي اشترى من اجله مثل ثمن الشراء وتكاليف الاعداد والتهيئة والمناولة والتزليل والتركيب والرسوم الجمركية الخ .

القيمة المتبقية (الخردة) Salvage:

هي القيمة التي تحصل عليها الشركة في نهاية العمر الإنتاجي للأصل وهي القيمة التي تقديرها للأصل كما لو كان اليوم خرده

العمر الإنتاجي (Life):

عبارة عن الفترة الزمنية التي يتوقع استخدام الاصل خلالها . او عدد الوحدات المنتجة او عدد الساعات ... الخ هناك اختلاف بين العمر الإنتاجي والمادي للأصل فقد يكون الاصل قادر على الاستمرار ولكن يتم استبعاده لعدة اسباب مثل التقادم او عدم مناسبة الاصل (مثلا مكان الانتاج اصبح غير كاف) او الحداثة مثل اجهزة الكمبيوتر.... الخ .

*تكلفة الاصل cost و القيمة المتبقية (الخردة) Salvage و العمر الإنتاجي Life هي نفس وسائط الدالة SLN وايضا كل دوال الإهلاك تستخدم نفس المصطلحات

* ذكرت المعايير الصادرة في ٢٠١٦ كلمة الاستهلاك بدلا من الاهلاك وهو مصطلح مايكروسوفت الذي تستخدمه

دوال الاهلاك الموجودة في الاكسيل

يحتوى برنامج MICROSOFT EXCEL على خمس دوال لطرق الاهلاك المختلفة وهما :

اسم الدالة	وصف الدالة	طريقة الإهلاك
SLN	Straight line	طريقة القسط الثابت
DB	Declining Balance	طريقة القسط المتناقص
DDB	Double Declining Balance	طريقة القسط المتناقص المضاعف
SYD	Sum Of Year's Digits	طريقة مجموع ارقام السنوات
VBD	Variable Declining Balance	طريقة القسط المتناقص المضاعف او طريقة اخرى

كل دوال الاهلاك موجودة ضمن مجموعة الدوال المالية في الاكسيل

يوجد دالتين إضافيتين لحساب الاهلاك للنظام المحاسبي الفرنسي فقط وليس الدولي وتم تناولهما

وسائط او مكونات دوال الاهلاك المستخدمة هي

اسم الوسيط	وصف
Cost	تكلفة الاصل
Salvage	قيمة الخردة
Life	العمر الإنتاجي
Period	فترة حساب الإهلاك
Month	عدد الاشهر في السنه الاولى . اذا اهمل يعتبر السنه الاولى ١٢ شهر
Factor	المعدل الذى يتناقص به الرصيد اذا اهمل يفترض انه طريقة DDB او معدل مضاعف
Start Period	فترة البداية
End Period	فترة النهاية
No Switch	التبديل الى طريقة القسط الثابت عندما يكون قسط الاهلاك اكبر من قسط اهلاك الرصيد المتناقص وهى قيمة منطقية TRUE FLASE

والان مع طرق الاهلاك بالتفصيل

اولا: طريقة القسط الثابت Straight line باستخدام الدالة SLN

شكل الدالة :

SLN(cost, salvage, life)

Cost مطلوبة. التكلفة الأولية للأصول.

Salvage مطلوبة. القيمة عند نهاية الإهلاك (الخرده او النفاية).

Life مطلوبة. عدد الفترات التي يتم فيها إهلاك الأصول (تسمى أحياناً فترة الانتفاع من الأصول).

تعتبر طريقة القسط الثابت من أكثر الطرق استخداماً في حساب الإهلاك حيث يتم توزيع تكلفة الأصل على عمره الإنتاجي وهنا

تتساوى قيمة اهتلاك لكل فتره مع الفترات الأخرى

وبموجب هذه الطريقة يجب الإهلاك كالاتي .

القيمة القابلة للاهتلاك = (تكلفة الاصل - الخردة) ÷ العمر الإنتاجي

ملاحظة:

يمكن ان تكون قيمة الخردة صفراً اذا كانت قيمة الاصل قليلة نسبياً ولكن اذا كان قيمة الاصل كبيره من قيمة الخردة ستكون مهمه

وجوهريه

مثال على طريقة القسط الثابت:

في ١ يناير ٢٠١٦ اشترت شركة الوطنية للدواجن آلة تقطيع بمبلغ ٢٢٠.٠٠٠ ج ويتوقع استخدامها في الإنتاج لمدة ٤ سنوات وتباع بعدها كخردة بمبلغ ٢٠.٠٠٠ ج

الحل

أولاً لابد من تحديد الآتي

220000 تكلفة الأصل (Cost)

20000 قيمة الخردة (Salvage)

4 العمر الإنتاجي (Life)

السنة	دالة SLN	FORMULATEXT	الطريقة اليدوية	FORMULATEXT
2010	50000	=SLN(\$E\$8,\$E\$9,\$E\$10)	50000	=(E8-E9)/4
2011	50000	=SLN(\$E\$8,\$E\$9,\$E\$10)	50000	=(E8-E9)/4
2012	50000	=SLN(\$E\$8,\$E\$9,\$E\$10)	50000	=(E8-E9)/4
2013	50000	=SLN(\$E\$8,\$E\$9,\$E\$10)	50000	=(E8-E9)/4
اجمالي	200000	=SUM(C13:C16)	200000	=SUM(E13:E16)

=SLN(\$E\$8;\$E\$9;\$E\$10)

Function Arguments

SLN

Cost: \$E\$8 = 220000

Salvage: \$E\$9 = 20000

Life: \$E\$10 = 4

= 50000

Returns the straight-line depreciation of an asset for one period.

Cost is the initial cost of the asset.

Formula result = 50000

[Help on this function](#)

OK Cancel

حساب القسط الثابت يدويا:

(تكلفة الأصل - قيمة الخردة) / العمر الإنتاجي أو (تكلفة الأصل - القيمة المتبقية) × نسبة الاهلاك

$$٥٠٠٠٠ = ٤ / (٢٠٠٠٠ - ٢٢٠٠٠٠) : \text{القسط الثابت السنوي}$$

كل سنة لها نفس قيمة القسط ٥٠٠٠٠ الف

ويمكن اعداد كشف الاهلاك على النحو التالي

السنة	تكلفة الاصل	الاهلاك السنوي للأصل	مجمع اهلاك الاصل	القيمة الدفترية للاصل
2010	220000	50000	50000	170000
2011	220000	50000	100000	120000
2012	220000	50000	150000	70000
2013	220000	50000	200000	20000

نلاحظ هنا ان القيمة الدفترية للأصل في نهاية الفترة عام ٢٠١٣ تساوى قيمة الخردة او النفاية

ملاحظة هامه

قد يكون يتم اهلاك الاصل بمعدل مثلا ٢٥% فكيف يتم تحويل هذه النسبة الى عمر او سنوات .

الحل: قسمة الرقم (١ ÷ ٢٥ = ٤ فترات)

ولكن للأسف دالة القسط الثابت . لا تحسب الا قسط سنة كاملة وهذا غير عملي لأنه قد يتم شراء او اقتناء الأصل في اى فترة خلال

السنة

كما في هذا المثال حساب الاهلاك من ٢٠١٠/٧/١ الى ٢٠١٥/١٢/٣١

الفترة	قسط الاهلاك
1	2100
2	4200
3	4200
4	4200
5	4200
6	2100
Total	21000

$$=IF(\$A8=1,SLN(\$A\$1,\$A\$2,\$A\$3)*((DATEDIF(\$A\$4,DATE(YEAR(\$A\$4),12,13),"m")+1)/12),$$

$$IF(\$A8=\$A\$3+1,$$

$$((12-(DATEDIF(\$A\$4,DATE(YEAR(\$A\$4),12,13),"m")+1))/12)*SLN(\$A\$1,\$A\$2,\$A\$3),SLN(\$A\$1,\$A\$2,\$A\$3)))$$

المعادلة المستخدمة :

```
=IF($A8=1,SLN($A$1,$A$2,$A$3)*((DATEDIF($A$4,DATE(YEAR($A$4),12,13),"m")+1)/12),  
IF($A8=$A$3+1,  
((12-  
(DATEDIF($A$4,DATE(YEAR($A$4),12,13),"m")+1)/12)*SLN($A$1,$A$2,$A$3),SLN($A$1,$A$2,$A$3)))
```

طبعا هذه معادلة ضخمة وأتمنى من مايكروسوفت ان تضع ذلك في الاعتبار بأشياء (٢) وسيط لفترة البداية والنهاية

(قدمت هذا الاقتراح لميكروسوفت) على الرابط التالي [اضغط على الرابط](#)

ويمكن أيضا حساب الفترة الأولى يدويا هكذا

```
=((11000-1000)/5)*6/12
```

وهكذا نكون قد انتهينا من طريقة الاهلاك الثابت ارجو ان اكون قد وفقت في توصيل المعلومة بشكل مناسب لكل المستويات.

طريقة القسط المتناقص (Accelerated Depreciation Method)

طرق الاهلاك المتناقص تقوم على فكرة تحميل الفترات الاولى من عمر الاصل بقيمة إهلاك عالية. ثم تبدأ قيمة الإهلاك بالتناقص في الفترات التالية.

ومن هذه الطرق في الاكسيل:

- طريقة القسط المتناقص (المضاعف او المعجل) باستخدام الدالة DDB
- طريقة القسط المتناقص الثابت باستخدام الدالة DB
- طريقة مجموع ارقام السنوات باستخدام الدالة SYD

دالة القسط المتناقص المضاعف DDB (Double-declining balance)

طبقاً لهذه الطريقة والتي تستخدم معدل اهلاك ضعف المعدل المستخدم في طريقة القسط الثابت ويظل هذا المعدل ثابتاً ويضرب في القيمة الدفترية المتناقصة من عام لآخر. وتستمر هذه العملية حتى تنخفض القيمة الدفترية للأصل الى قيمة الخردة المقدرة له وهنا يتوقف احتساب الاهلاك. وعلى عكس الطرق الأخرى فإنه لا تطرح قيمة الخردة عند حساب تكلفة الاهلاك.

شكل الدالة : DDB(cost, salvage, life, period, [factor])

Function Arguments ? X

DDB

Cost	التكلفة	↑	= number
Salvage	الخردة	↑	= number
Life	العمر الانتاجي	↑	= number
Period	الفترة	↑	= number
Factor	معامل: اذا ترك فارغ يفترض انها قسط مزدوج	↑	= number

=

Returns the depreciation of an asset for a specified period using the double-declining balance method or some other method you specify.

Cost is the initial cost of the asset.

Formula result =

[Help on this function](#) OK Cancel

Cost مطلوبة. التكلفة الأولية للأصول.

Salvage مطلوبة. القيمة عند نهاية الإهلاك (وتسمى في بعض الأحيان القيمة الباقية للأصول).

Life مطلوبة. عدد الفترات التي يتم فيها إهلاك الأصل (تسمى أحياناً العمر الإنتاجي للأصل).

Period مطلوبة. الفترة التي تريد حساب الإهلاك فيها

Factor اختيارية. إذا تم حذف الوسيطة factor، فسيقتضئ أنها 2 (أسلوب الاستهلاك المتناقص المزدوج).

ملاحظه

المعامل Factor : في الحياه العملية تستخدم العديد من مضاعفات القسط الثابت كان يكون المعدل ٢٠٠% او ١٥٠% من القسط الثابت ويمكن أيضا استبدال النسب المئوية بأعداد صحيحة ١ او ٢

وإذا اهل هذا الوسيط يفترض انها ٢ او ٢٠٠% أو أسلوب الاهلاك المضاعف او المزدوج

وللتوصل إلى قسط الإهلاك السنوي فلا بد من إتباع الخطوات التالية :

(١) حساب نسبة معدل الاهلاك وفقا لطريقة القسط الثابت ثم مضاعفة هذه النسبة

(١٠٠% ÷ العمر الإنتاجي) × ٢ او (٢ ÷ العمر الإنتاجي)

(٢) حساب مصروف الإهلاك السنوي

القيمة الدفترية للأصل في بداية الفترة (التكلفة - مجمع الإهلاك) × نسبة معدل الإهلاك

انتبه ملاحظات هامه

- قسط إهلاك السنة الأخيرة تبعا لهذه الطريقة تتمثل بالقيمة الدفترية للسنة الأخيرة مطروحا منها قيمة الخردة (القيمة الدفترية-الخردة)
- في الممارسات العملية فانه تستخدم العديد من مضاعفات القسط الثابت كأن يكون المعدل ٢٠٠% او ١٥٠% او اعداد صحيحة ١ او ٢ الخ.

مثال عملي (استخدام طريقة القسط المتناقص في حساب الاهلاك) باستخدام الدالة DDB

في ١ يناير ٢٠١٠ اشترت شركة الوطنية للدواجن آلة تقطيع بمبلغ ٢٢٠٠٠٠ ج ويتوقع استخدامها في الانتاج لمدة ٤ سنوات وتباع بعدها كخردة بمبلغ ٢٠٠٠٠ ج

والمطلوب منك هو:

تحديد مصروف الاهتلاك السنوي وفق طريقة القسط المتناقص يدويا
وباستخدام الدالة DDB (Double Declining Balance)

الحل : باستخدام الطريقة اليدوية

السنة	القيمة الدفترية للأصل في أول الفترة	نسبة الاستهلاك (١)*	مصرف الاستهلاك*(٢)	مجمع الاستهلاك	القيمة الدفترية آخر الفترة
0					220000
1	220000	50%	110000	110000	110000
2	110000	50%	55000	165000	55000
3	55000	50%	27500	192500	27500
4	27500	50%	7500	200000	20000

ملاحظات على الحل :

- تم حساب نسبة الاهلاك كما ذكر سابقا (٢÷العمر الإنتاجي)
- مصرف الاهلاك (القيمة الدفترية في بداية الفترة * نسبة الاستهلاك)
- في الفترة الأخيرة يتوقف احتساب الاهلاك عندما تساوى القيمة الدفترية قيمة الخردة = القيمة الدفترية في بداية الفترة - قيمة الخردة . (٢٠٠٠٠-٢٧٥٠٠).

الحل : باستخدام الدالة DDB

	D	C	B	A	
9	220000				تكلفة الاصل (Cost)
10	20000				قيمة الخردة (Salvage)
11	4				العمر الانتاجي (Life)
12	FORMULATEXT	دالة DDB	الفترة	السنة	
13	=DDB(\$D\$9,\$D\$10,\$D\$11,\$B13,200%)	110000	1	2010	
14	=DDB(\$D\$9,\$D\$10,\$D\$11,\$B14,200%)	55000	2	2011	
15	=DDB(\$D\$9,\$D\$10,\$D\$11,\$B15,200%)	27500	3	2012	
16	=DDB(\$D\$9,\$D\$10,\$D\$11,\$B16,200%)	7500	4	2013	
17		200000			اجمالي
18					

Function Arguments ? X

DDB

Cost	=\$D\$9	=	220000	
Salvage	=\$D\$10	=	20000	
Life	=\$D\$11	=	4	
Period	=\$B\$13	=	1	
Factor	=200%	=	2	

= 110000

المعامل يستخدم العديد من مضاعفات القسط الثابت في الحياة العملية كان يكون المعدل 200 % او 150% من معدل القسط الثابت ويمكن استبدال النسب المئوية باعداد صحيحة 1 او 2 . واذا اهمل هذا الوسيط يفترض انها 2 او 200% يعنى اذا ترك فارغ .

Formula result = 110000

[Help on this function](#) OK Cancel

ملاحظة :

ذكر موقع مايكروسوفت طريقة عمل هذه الدالة يدويا

يحسب أسلوب الاستهلاك المتناقص المزدوج الاستهلاك بمعدل متزايد. يكون الإهلاك في أعلى معدلاته في الفترة الأولى ثم ينخفض في الفترات المتعاقبة. تستخدم DDB الصيغة التالية لحساب الإهلاك في فترة معينة:

$$\text{Min}((\text{cost} - \text{total depreciation from prior periods}) * (\text{factor}/\text{life}), (\text{cost} - \text{salvage} - \text{total depreciation from prior periods}))$$

C5 : X ✓ fx =MIN((\$D\$1-SUM(C\$4:C4))*(2/\$D\$3),(\$D\$1-\$D\$2-SUM(C\$4:C4)))				
	A	B	C	D
1	تكلفة الاصل (Cost)			220000
2	قيمة الخردة (Salvage)			20000
3	العمر الانتاجي (Life)			4
4	السنة	الفترة	دالة DDB	FORMULATEXT
5	2010	1	110000	=MIN((\$D\$1-SUM(C\$4:C4))*(2/\$D\$3),(\$D\$1-\$D\$2-SUM(C\$4:C4)))
6	2011	2	55000	=MIN((\$D\$1-SUM(C\$4:C5))*(2/\$D\$3),(\$D\$1-\$D\$2-SUM(C\$4:C5)))
7	2012	3	27500	=MIN((\$D\$1-SUM(C\$4:C6))*(2/\$D\$3),(\$D\$1-\$D\$2-SUM(C\$4:C6)))
8	2013	4	7500	=MIN((\$D\$1-SUM(C\$4:C7))*(2/\$D\$3),(\$D\$1-\$D\$2-SUM(C\$4:C7)))
9	اجمالي		0	
10	Min((cost - total depreciation from prior periods) * (factor/life), (cost - salvage - total depreciation from prior periods))			

المعادلة المستخدمة:

$$=MIN((\$D\$1-SUM(C\$4:C4))*(2/\$D\$3),(\$D\$1-\$D\$2-SUM(C\$4:C4)))$$

ولكن أيضا تظل نفس المشكلة وهي اهلاك اصل خلال فترة معينة من السنه وليس سنة كاملة ولا تحلها الدالة DDB في الاكسيل

مثال: في ٢٠١٠/١٠/١ قامت الشركة بشراء الة جديده قيمتها ٢٢٠٠٠٠ ويتوقع استخدامها في الإنتاج ٤ سنوات وتباع بعدها كخردة بمبلغ ٢٠٠٠.

المطلوب: حساب الاهلاك المتناقص المضاعف للأعوام ٢٠١٠ و ٢٠١١ و ٢٠١٢ و ٢٠١٣ و ٢٠١٤

الحل:

١- مصروف اهلاك عام ٢٠١٠ عن (٣ شهور) عباره عن $2 \times (4 \div 22000) = 11000 \times (3 \div 12) = 2750$

٢- اما مصروف اهلاك عام ٢٠١١ يتم تقسيمه الى قسمين:

القسم الأول: اهلاك ٩ شهور من العام الأول للأصل $8250 = ((12 \div 9) \times 11000)$

القسم الثاني: اهلاك ٣ شهور من العام التالي للأصل $13750 = ((12 \div 3) \times 5500)$

إجمالي اهلاك عام ٢٠١١ $96250 = (8250 + 13750)$

وهكذا

توزيع اهلاك السنة الأولى



حتى انتهاء ٢٠١٠			حتى انتهاء ٢٠١١								
10	11	12	1	2	3	4	5	6	7	8	9
11000 × (3/12)			11000 × (9/12)								

السنة المالية	الاهلاك السنه (١) ٢٠١٠	الاهلاك السنه (٢) ٢٠١١	الاهلاك السنه (٣) ٢٠١٢	الاهلاك السنه (٤) ٢٠١٣	الاهلاك السنه (٥) ٢٠١٤	مجمع الاهلاك
الفترة الأولى	2,750.00	8,250.00	4,125.00	2,062.50	562.50	20,000.00
الفترة الثانية	0	1375	687.5	187.5	0	
الفترة الأولى	لا يوجد	5500 × (3 ÷ 12)	2750 × (3 ÷ 12)	750 × (3 ÷ 12)	لا يوجد	
الفترة الأولى	11000 × (3 ÷ 12)	5500 × (9 ÷ 12)	5500 × (9 ÷ 12)	2750 × (9 ÷ 12)	750 × (9 ÷ 12)	
الفترة الثانية	لا يوجد	لا يوجد	لا يوجد	لا يوجد	لا يوجد	

الحل باستخدام الأكسيل :

الفترة	قسط الإهلاك	النتيجة	اهلاك الفترة الأولى	النتيجة	اهلاك الفترة الثانية	المجموع	العام
1	2750	2,750.00	11000 × (3÷12)	0.00	لا يوجد	2,750.00	اهلاك السنة (1) ٢٠١٠
2	9625	8,250.00	11000 × (9÷12)	1,375.00	5500 × (3÷12)	9,625.00	اهلاك السنة (٢) ٢٠١١
3	4812.5	4,125.00	5500 × (9÷12)	687.50	2750 × (3÷12)	4,812.50	اهلاك السنة (٣) ٢٠١٢
4	2250	2,062.50	2750 × (9÷12)	187.50	750 × (3÷12)	2,250.00	اهلاك السنة (٤) ٢٠١٣
5	562.5	562.50	750 × (9÷12)	0.00	0.00	562.50	اهلاك السنة (٥) ٢٠١٤
Total	20000					20,000.00	مجموع الاهلاك

مصدر المثال : كتيب معايير المحاسبة الدولية لاعداد التقارير المالية IFRS EXPERT

المعادلة المستخدمة :

$$=SUM(IF(A8=1,ROUND(DDB(\$A\$1,\$A\$2,\$A\$3,A8)*((DATEDIF(\$A\$4,DATE(YEAR(\$A\$4),12,31),"M")+1)/12),0),IFERROR(DDB(\$A\$1,\$A\$2,\$A\$3,\$A7)*(DATEDIF(DATE(YEAR(\$A\$4),1,1),DATE(YEAR(\$A\$4),MONTH(\$A\$4),DAY(\$A\$4)),"M")/12),0)),IF(A8>1,IFERROR(DDB(\$A\$1,\$A\$2,\$A\$3,\$A9-1)*((12-(DATEDIF(DATE(YEAR(\$A\$4),1,1),DATE(YEAR(\$A\$4),MONTH(\$A\$4),DAY(\$A\$4)),"M")))/12),0),0))$$

تحليل المعادلة السابقة:

معادلة اهلاك عام ٢٠١٠:

$$=IFERROR(ROUND(DDB(\$A\$1,\$A\$2,\$A\$3,A8)*((DATEDIF(\$A\$4,DATE(YEAR(\$A\$4),12,31),"M")+1)/12),0),0)$$

وتحلل هذه المعادلة كالتالي :

مصروف اهلاك عام ٢٠١٠ عن (٣ شهور) عبارة عن $2 \times (4 \div 12) = 0.6666666666666667$ عن $11000 \times (0.6666666666666667) = 7333.333333333333$

معادلة اهلاك عام ٢٠١١ وينقسم الى قسمين

القسم الأول : اهلاك ٩ شهور من العام الأول للأصل $8250 = ((12 \div 9) \times 11000)$

$$=IF(A9=1,ROUND(DDB(\$A\$1,\$A\$2,\$A\$3,A9)*((DATEDIF(\$A\$4,DATE(YEAR(\$A\$4),12,31),"M")+1)/12),0),IFERROR(DDB(\$A\$1,\$A\$2,\$A\$3,A8)*(DATEDIF(DATE(YEAR(\$A\$4),1,1),DATE(YEAR(\$A\$4),MONTH(\$A\$4),DAY(\$A\$4)),"M")/12),0))$$

القسم الثاني : اهلاك ٣ شهور من العام التالي للأصل $1375 = ((12 \div 3) \times 5500)$

$$=IF(A9>1,IFERROR(DDB(\$A\$1,\$A\$2,\$A\$3,\$A10-1)*((12-(DATEDIF(DATE(YEAR(\$A\$4),1,1),DATE(YEAR(\$A\$4),MONTH(\$A\$4),DAY(\$A\$4)),"M")))/12),0),0)$$

إجمالي اهلاك عام ٢٠١١ = (١٣٧٥ + ٨٢٥٠) = ٩٦٢٥

=SUM(IF(A9=1,ROUND(DDB(\$A\$1,\$A\$2,\$A\$3,A9)*((DATEDIF(\$A\$4,DATE(YEAR(\$A\$4),12,31),"M")+1)/12),0),IFERROR(DDB(\$A\$1,\$A\$2,\$A\$3,\$A8)*(DATEDIF(DATE(YEAR(\$A\$4),1,1),DATE(YEAR(\$A\$4),M ONTH(\$A\$4),DAY(\$A\$4)),"M")/12),0)), IF(A9>1,IFERROR(DDB(\$A\$1,\$A\$2,\$A\$3,\$A10-1)*((12-(DATEDIF(DATE(YEAR(\$A\$4),1,1),DATE(YEAR(\$A\$4),MONTH(\$A\$4),DAY(\$A\$4)),"M")))/12),0),0))

وهكذا مع السنوات التالية:

جدول الاهلاك

السنة	القيمة الدفترية للاصل في اول الفتره	مصروف الاستهلاك*(٢)	مجمع الاستهلاك	القيمة الدفترية اخر الفتره
0	22000			22000
1	22000	2,750.00	2750	19250
2	19250	9,625.00	12375	9625
3	9625	4,812.50	17187.5	4812.5
4	4812.5	2,250.00	19437.5	2562.5
5	2562.5	562.50	20000	2000

وهكذا نكون قد حاولنا في حل هذه المشكلة وهي الفترات الجزئية والتي لا تقوم دوال الاهلاك بمفردها بذلك

طريقة مجموع ارقام السنوات باستخدام الدالة SYD(Sum Of Year's Digits)

بموجب هذه الطريقة فان مصروف اهلاك السنة الأولى ثم يبدأ بالتدرج بالتناقص ولكن بدرجة اقل من طريقة القسط المضاعف
شكل الدالة :

SYD(cost, salvage, life, per)

Function Arguments ? X

SYD

Cost	التكلفة	↑		= number
Salvage	الخرده	↑		= number
Life	العمر الانتاجي	↑		= number
Per	فترة الاهلاك	↑		= number

=

Returns the sum-of-years' digits depreciation of an asset for a specified period.

Cost is the initial cost of the asset.

Formula result =

[Help on this function](#) OK Cancel

ويتم حساب هذه الدالة كما يلي :

$$SYD = \frac{(cost - salvage) * (life - per + 1) * 2}{(life)(life + 1)}$$

مثال عملي : (استخدام طريقة مجموع ارقام السنوات في حساب الاهلاك) باستخدام الدالة SYD

في ١ يناير ٢٠١٠ اشترت شركة الوطنية للدواجن آلة تقطيع بمبلغ ٢٢٠.٠٠٠ ج ويتوقع استخدامها في الانتاج لمدة ٤ سنوات وتباع بعدها كخرده بمبلغ ٢٠.٠٠٠ ج والمطلوب منك هو:
تحديد مصروف الإهلاك السنوي وفق طريقة مجموع ارقام السنوات يدويا وباستخدام الدالة SYD (Sum Of Year's Digits)

D	C	B	A	
220000				3 تكلفة الاصل (Cost)
20000				4 قيمة الخردة (Salvage)
4				5 العمر الانتاجي (Life)
10				6 مجموع ارقام السنوات
FORMULATEXT	دالة SYD	الفترة	السنة	7
=SYD(\$D\$3,\$D\$4,\$D\$5,B8)	80000	1	2010	8
=SYD(\$D\$3,\$D\$4,\$D\$5,B9)	60000	2	2011	9
=SYD(\$D\$3,\$D\$4,\$D\$5,B10)	40000	3	2012	10
=SYD(\$D\$3,\$D\$4,\$D\$5,B11)	20000	4	2013	11
	200000			12 اجمالي مصروف الاهلاك

Function Arguments ? X

SYD

Cost: \$D\$3 = 220000 التكلفة

Salvage: \$D\$4 = 20000 الخردة

Life: \$D\$5 = 4 العمر الانتاجي

Per: B8 = 1 فترة الاهلاك

= 80000

Returns the sum-of-years' digits depreciation of an asset for a specified period.
Cost is the initial cost of the asset.

Formula result = 80,000.00 ج.م.

[Help on this function](#) OK Cancel

الطريقة التالية اليدوية طبقا لما ورد من مايكروسوفت

A	B	C	D	E
(Cost) تكلفة الاصل	220000			
(Salvage) قيمة الخردة	20000			
(Life) العمر الانتاجي	4			
السنة	الفترة	دالة SYD	FORMULATEXT	
2010	1	80000	=(((\$B\$1-\$B\$2)*(\$B\$3-\$B6+1)*2)/(((\$B\$3*(\$B\$3+1))))	
2011	2	60000	=(((\$B\$1-\$B\$2)*(\$B\$3-\$B7+1)*2)/(((\$B\$3*(\$B\$3+1))))	
2012	3	40000	=(((\$B\$1-\$B\$2)*(\$B\$3-\$B8+1)*2)/(((\$B\$3*(\$B\$3+1))))	
2013	4	20000	=(((\$B\$1-\$B\$2)*(\$B\$3-\$B9+1)*2)/(((\$B\$3*(\$B\$3+1))))	
اجمالي مصروف الاهلاك		200000		

طريقة أخرى :

ويتم احتساب مصروف الإهلاك بموجب هذه الطريقة كما يلي :

١. تحديد القيمة القابلة للإهلاك = (التكلفة - الخردة)

٢. حساب مجموع ارقام السنوات مثلا (١+٢+٣+٤) = ١٠.

او مجموع ارقام السنين = $\frac{2}{(1+n)}$ حيث ن = عدد سنوات العمر الإنتاجي

٣. ضرب (١ خطوه × ٢ خطوه) ضرب القيمة القابلة للإهلاك × (الباقى من عمر الأصل / مجموع ارقام السنوات)

I	H	G	F	E	D	C	B	A	
	القيمة الدفترية	مجمع الاهلاك	مصروف الاهلاك	معدل الاهلاك (القيمة المتبقية من العمر الانتاجي / مجموع ارقام السنوات)	الباقى من عمر الأصل	القيمة القابلة للاهتلاك	السنة		
	220000							0	17
	140000	80000	80000	4/10	4	200000	1	18	
	80000	140000	60000	3/10	3	200000	2	19	
	40000	180000	40000	2/10	2	200000	3	20	
	20000	200000	20000	1/10	1	200000	4	21	
			200000						22
						اجمالي مصروف الاهلاك			23

المعادلات المستخدمة

H	G	E	D	C	B	A	
القيمة الدفترية	مجمع الاهلاك	مصروف الاهلاك	معدل الاهلاك (القيمة المتبقية من القيمة المتبقية من العمر الانتاجي / مجموع ارقام السنوات)	الباقى من عمر الأصل	القيمة القابلة للاهتلاك	السنة	
=\$D\$3-G18						0	17
=\$D\$3-G19	=SUM(\$E\$7:E1	=B19*(\$C19/\$D\$6)	4/10	=(D\$5-A19)+1	=(D\$3-D\$4)	1	18
=\$D\$3-G20	=SUM(\$E\$7:E2	=B20*(\$C20/\$D\$6)	3/10	=(D\$5-A20)+1	=(D\$3-D\$4)	2	19
=\$D\$3-G21	=SUM(\$E\$7:E2	=B21*(\$C21/\$D\$6)	2/10	=(D\$5-A21)+1	=(D\$3-D\$4)	3	20
=\$D\$3-G22	=SUM(\$E\$7:E2	=B22*(\$C22/\$D\$6)	1/10	=(D\$5-A22)+1	=(D\$3-D\$4)	4	21
		=SUM(E19:F22)			اجمالي مصروف الاهلاك		22
							23

ملاحظات على الحل :

مصروف الاهلاك عبارته عن : القيمة القابلة للإهلاك (التكلفة - الخردة) = (٢٠٠٠٠ - ٢٢٠٠٠) = ٢٠٠٠٠

مضروب في الباقي من عمر الأصل مقسوم ١٠/٤ مجموع ارقام السنوات = (١+٢+٣+٤)

الفترة الأولى = $\frac{10}{4} \times 20000 = 80000$ الفترة الثانية = $\frac{10}{3} \times 20000 = 60000$

والفترة الثالثة = $\frac{10}{2} \times 20000 = 40000$ والفترة الأخيرة = $\frac{10}{1} \times 20000 = 20000$

الاجمالي = (٨٠٠٠٠ + ٦٠٠٠٠ + ٤٠٠٠٠ + ٢٠٠٠٠) = ٢٠٠٠٠٠ اجمالي مصروف الاهلاك

حساب قسط الاهلاك على فترة جزئية باستخدام طريقة مجموع ارقام السنوات

مثال: في ١/١٠/٢٠٠٩ قامت الشركة بشراء اصل تكلفته ٦٤٠٠٠ ج ويتوقع ان يستخدم لمدة ٥ سنوات يباع بعدها كخردة بمبلغ ٤٠٠٠

هنا يحسب مصروف الاهلاك كالتالى

السنة المالية	قسط الاهلاك		
اهلاك عام ٢٠٠٩	الفترة الأولى	$20000 \times (3 \div 12)$	5,000.00
	الفترة الثانية	لا يوجد	لا يوجد
اهلاك عام ٢٠١٠	الفترة الأولى	$20000 \times (9 \div 12)$	15,000.00
	الفترة الثانية	$16000 \times (3 \div 12)$	4,000.00
اهلاك عام ٢٠١١	الفترة الأولى	$16000 \times (9 \div 12)$	12,000.00
	الفترة الثانية	$12000 \times (3 \div 12)$	3,000.00
اهلاك عام ٢٠١٢	الفترة الأولى	$12000 \times (9 \div 12)$	9,000.00
	الفترة الثانية	$8000 \times (3 \div 12)$	2,000.00
اهلاك عام ٢٠١٣	الفترة الأولى	$8000 \times (9 \div 12)$	6,000.00
	الفترة الثانية	$4000 \times (3 \div 12)$	1,000.00
اهلاك عام ٢٠١٣	الفترة الأولى	$4000 \times (9 \div 12)$	3,000.00
	الفترة الثانية	لا يوجد	لا يوجد
مجمع الاهلاك	60,000.00		

ويكون جدول الاهلاك كالتالى :

السنة	القيمة الدفترية للاصل في اول الفتره	مصروف الاستهلاك*(٣)	مجمع الاستهلاك	القيمة الدفترية اخر الفتره
0	64000			64000
1	64000	5,000.00	5000	59000
2	59000	19,000.00	24000	40000
3	40000	15,000.00	39000	25000
4	25000	11,000.00	50000	14000
5	14000	7,000.00	57000	7000
6	7000	3,000.00	60000	4000
المجموع		60000		

وقد تم حساب الجدول أعلاه باستخدام معادلات الأكسيل كالتالي :

الفترة	قسط الإهلاك	النتيجة	اهلاك الفترة الأولى	النتيجة	اهلاك الفترة الثانية	المجموع	العام
1	5000	5,000.00	20000 × (3÷12)	0.00	لا يوجد	5,000.00	اهلاك عام ٢٠٠٩
2	19000	15,000.00	20000 × (9÷12)	4,000.00	16000 × (3÷12)	19,000.00	اهلاك عام ٢٠١٠
3	15000	12,000.00	16000 × (9÷12)	3,000.00	12000 × (3÷12)	15,000.00	اهلاك عام ٢٠١١
4	11000	9,000.00	12000 × (9÷12)	2,000.00	8000 × (3÷12)	11,000.00	اهلاك عام ٢٠١٢
5	7000	6,000.00	8000 × (9÷12)	1,000.00	4000 × (3÷12)	7,000.00	اهلاك عام ٢٠١٣
6	3000	3,000.00	4000 × (9÷12)	0.00	لا يوجد	3,000.00	اهلاك عام ٢٠١٤
Total	60000					60,000.00	مجموع الأهلاك

مصدر المثال : هيبر معايير المحاسبة الدولية لاعداد التقارير المالية IFRS EXPERT

المعادلة المستخدمة :

$$=SUM(IF(A8=1,ROUND(SYD(A1,A2,A3,A8)*((DATEDIF(A4,DATE(YEAR(A4),12,31),"M")+1)/12),0),IFERROR(SYD(A1,A2,A3,$A7)*(DATEDIF(DATE(YEAR($A$4),1,1),DATE(YEAR($A$4),MONTH($A$4),DAY($A$4)),"M")/12),0)),IF(A8>1,IFERROR(SYD(A1,A2,A3,$A9-1)*((12-DATEDIF(DATE(YEAR(A4),1,1),DATE(YEAR(A4),MONTH(A4),DAY(A4)),"M"))/12),0),0))$$

تحليل المعادلة السابقة:

معادلة اهلاك عام ٢٠٠٩ :

$$=IF(A8=1,ROUND(SYD(A1,A2,A3,A8)*((DATEDIF(A4,DATE(YEAR(A4),12,31),"M")+1)/12),0),IFERROR(SYD(A1,A2,A3,$A7)*(DATEDIF(DATE(YEAR($A$4),1,1),DATE(YEAR($A$4),MONTH($A$4),DAY($A$4)),"M")/12),0))$$

مصروف اهلاك عام ٢٠٠٩ عن (٣ شهور) عبارة عن

(القيمة القابلة للإهلاك × معدل الاهلاك السنوي (مجموع ارقام السنوات)) × (١٢/٣ شهر)

$$0 \dots = (12 \div 3) \times 20000 = (15/5) * (4000 - 16000) =$$

مجموع ارقام السنوات ١٥ عبارة عن

$$15 = 5 + 4 + 3 + 2 + 1$$

معادلة اهلاك عام ٢٠١١ وينقسم الى قسمين

القسم الأول : اهلاك ٩ شهور من العام الأول للأصل $15000 = ((12 \div 9) \times 20000)$

=IF(A9=1,ROUND(SYD(\$A\$1,\$A\$2,\$A\$3,A9)*((DATEDIF(\$A\$4,DATE(YEAR(\$A\$4),12,31),"M")+1)/12),0),IFERROR(SYD(\$A\$1,\$A\$2,\$A\$3,\$A8)*(DATEDIF(DATE(YEAR(\$A\$4),1,1),DATE(YEAR(\$A\$4),MONTH(\$A\$4),DAY(\$A\$4)),"M")/12),0))

القسم الثاني : اهلاك ٣ شهور من العام التالي للأصل $4000 = ((12 \div 3) \times 16000)$

=IF(A9>1,IFERROR(SYD(\$A\$1,\$A\$2,\$A\$3,\$A10-1)*((12-DATEDIF(DATE(YEAR(\$A\$4),1,1),DATE(YEAR(\$A\$4),MONTH(\$A\$4),DAY(\$A\$4)),"M"))/12),0),0)

إجمالي اهلاك عام ٢٠١١ $19000 = (4000 + 15000) = 2011$

=SUM(IF(A9=1,ROUND(DDB(\$A\$1,\$A\$2,\$A\$3,A9)*((DATEDIF(\$A\$4,DATE(YEAR(\$A\$4),12,31),"M")+1)/12),0),IFERROR(DDB(\$A\$1,\$A\$2,\$A\$3,\$A8)*(DATEDIF(DATE(YEAR(\$A\$4),1,1),DATE(YEAR(\$A\$4),MONTH(\$A\$4),DAY(\$A\$4)),"M")/12),0)),IF(A9>1,IFERROR(DDB(\$A\$1,\$A\$2,\$A\$3,\$A10-1)*((12-DATEDIF(DATE(YEAR(\$A\$4),1,1),DATE(YEAR(\$A\$4),MONTH(\$A\$4),DAY(\$A\$4)),"M"))/12),0),0))

وهكذا مع السنوات التالية

يوجد طريقة أخرى من طرق الاهلاك المتناقص تسمى طريقة وحدات النشاط ولكن لا يوجد لها دالة في الاكسيل

طريقة وحدات الإنتاج او النشاط:

تتبع هذه الطريقة عندما يمكن تقدير عدد وحدات الإنتاج التي يمكن ان تتحقق على مدى حياة الأصل ويحسب اهلاك كل فتره على أساس مستوى انتاج الفترة. ولكن تحديد عدد الوحدات امر صعب ويخضع للاجتهادات الشخصية وتلائم هذه الطريقة السيارات الطائرات وبعض الأصول الأخرى التي يسهل تقدير وحدات انتاجها.

وللتوصل الى مصروف الاهلاك السنوي يتبع الخطوات التالية:

١ تقدير عدد وحدات الإنتاج خلال العمر الانتاجي مثلا السيارة ٢٠ الف كم خلال ٥ سنوات

٢ تحديد القيمة القابلة للاهلاك (تكلفة الأصل- الخردة) (٤٠٠٠-٥٤٠٠٠)=٥٠٠٠٠

٣ حساب معدل الاهلاك لكل وحده (القيمة القابلة للاهلاك ÷ عدد وحدات الإنتاج) وهذا المعدل ثابت على مدى العمر الانتاجي

$$0.25 = 20000 / 80000$$

٤ حساب مصروف الاهلاك السنوي = عدد الوحدات الفعلية × معدل الاهلاك ٤٠٠٠ × ٢٥%

وهكذا

مثال: في ٢٠٠٩ تم شراء سيارة بمبلغ ٥٤٠٠٠ ومقدر عمرها الإنتاجي ٤ سنوات وقيمة الخردة ٤٠٠٠ ويتوقع ان تقطع مسافة ٢٠٠ الف

كم خلال العمر الإنتاجي لها .

احسب قسط الاهلاك السنوي بطريقة وحدات الإنتاج

	A	B	C	D
1	تكلفة الأصل	54000		
2	الخردة	4000		
3	العمر الانتاجي	4		
4	عدد الوحدات	200000		
5				
6	العام	المسافة الفعلية	معدل الاهلاك	مصروف الاهلاك
7	2009	40000 KM	25%	10000
8	2010	60000 KM	25%	15000
9	2011	30000 KM	25%	7500
10	2012	50000 KM	25%	12500
11	2013	20000 KM	25%	5000
12	Total	200000 KM		50000
13				
14			=(54000-4000)/200000	=40000*0.25
15				
16				

طريقة القسط المتناقص الثابت باستخدام الدالة DB

تُرجع هذه الدالة إهلاك أصول لفترة معينة باستخدام أسلوب الرصيد المتناقص الثابت وبمعدل ثابت في التناقص سنويا

DB(cost, salvage, life, period, [month])

Function Arguments ? X

DB

Cost	التكلفة	↑	= number
Salvage	الخرده	↑	= number
Life	العمر الانتاجي	↑	= number
Period	الفترة	↑	= number
Month	عدد الاشهر فى السنة الاولى	↑	= number

=

Returns the depreciation of an asset for a specified period using the fixed-declining balance method.

Cost is the initial cost of the asset.

Formula result =

[Help on this function](#) OK Cancel

Cost تكلفة الأصل

Salvage القيمة عند نهاية الإهلاك (تسمى في بعض الأحيان العمر الإنتاجي للأصل).

Life عدد الفترات التي يتم فيها إهلاك الأصل (تسمى أحيانا العمر الإنتاجي للأصل).

Period مطلوبة. الفترة التي تريد حساب الإهلاك فيها

Month اختيارية. عدد الأشهر في السنة الأولى. في حال حذف الوسيطة month، يفترض أنها تساوي ١٢.

تستخدم DB الصيغ التالية لحساب الإهلاك لفترة معينة:

(التكلفة - الإهلاك الكلي من فترات سابقة او التراكمي) * معدل الإهلاك وبحسب المعدل كالتالي :

المعدل = ١ - ((القيمة الباقية / التكلفة) ^ (١ / العمر الإنتاجي))، مقرباً إلى ثلاث منازل عشرية

يعد الإهلاك للفترتين الأولى والأخيرة حالة خاصة

للفترة الأولى: التكلفة * المعدل * الشهر / ١٢

للفترة الأخيرة: ((التكلفة - الإهلاك الكلي من فترات سابقة) * المعدل * (١٢ - الشهر)) / ١٢

المثال التالي يوضح طريقة عمل الدالة DB

C	B	A	
1000000	(Cost) تكلفة الاصل		1
100000	(Salvage) قيمة الخرده		2
6	(Life) العمر الانتاجى		3
	عدد الأشهر (تاريخ الش		4
			5
قسط الاهلاك باستخدام DB		السنة	
=DB(\$C\$1,\$C\$2,\$C\$3,A7,7)	186083.33	1	7
=DB(\$C\$1,\$C\$2,\$C\$3,A8,7)	259639.42	2	8
=DB(\$C\$1,\$C\$2,\$C\$3,A9,7)	176814.44	3	9
=DB(\$C\$1,\$C\$2,\$C\$3,A10,7)	120410.64	4	10
=DB(\$C\$1,\$C\$2,\$C\$3,A11,7)	81999.64	5	11
=DB(\$C\$1,\$C\$2,\$C\$3,A12,7)	55841.76	6	12
=DB(\$C\$1,\$C\$2,\$C\$3,A13,7)	15845.10	7	13
=SUM(B7:B13)	896634.33		14

Function Arguments

DB

Cost	<input type="text" value="\$C\$1"/>	↑	= 1000000
Salvage	<input type="text" value="\$C\$2"/>	↑	= 100000
Life	<input type="text" value="\$C\$3"/>	↑	= 6
Period	<input type="text" value="A7"/>	↑	= 1
Month	<input type="text" value="7"/>	↑	= 7

= 186083.3333

Returns the depreciation of an asset for a specified period using the fixed-declining balance method.

Cost is the initial cost of the asset.

Formula result = 186083.33

[Help on this function](#)

الطريقة العادية :

تستخدم DB الصبغ التالية لحساب الإهلاك لفترة معينة:

(التكلفة - الإهلاك الكلي من فترات سابقة او التراكمي) * معدل الإهلاك ويحسب المعدل كالتالي :

المعدل = ١ - ((القيمة الباقية / التكلفة) ^ (١ / العمر الإنتاجي))، مقرباً إلى ثلاث منازل عشرية

ويكون المعدل هكذا :

$$=1-((100000/1000000)^(1/6)) = 31.9\%$$

يعد الإهلاك للفترتين الأولى والأخيرة حالة خاصة

للفترة الأولى: التكلفة * المعدل * الشهر / ١٢

$$=1000000*31.9\%*(7/12)$$

للفترة الأخيرة: ((التكلفة - الإهلاك الكلي من فترات سابقة) * المعدل * (١٢ - الشهر)) / ١٢

$$=1000000*31.9\%*(5/12)$$

السنة	قسط الإهلاك السنوي	ملاحظات	مجمع الإهلاك	القيمة الدفترية للاصل (القيمة الباقية)
0			0	1000000
1	186083.33	الإهلاك في السنة الأولى، مع احتساب ٧ أشهر فقط	186083	813917
2	259639.42	الإهلاك في السنة الثانية	445723	554277
3	176814.44	الإهلاك في السنة الثالثة	622537	377463
4	120410.64	الإهلاك في السنة الرابعة	742948	257052
5	81999.64	الإهلاك في السنة الخامسة	824947	175053
6	55841.76	الإهلاك في السنة السادسة	880789	119211
7	15845.10	الإهلاك في السنة السابعة، مع احتساب ٥ أشهر فقط	896634	103366
	896634.326	الإجمالي		

الدالة VDB (variable declining balance) الرصيد المتناقص المتغير

تُرجع هذه الدالة استهلاك أحد الأصول لأي فترة تحددها، بما فيها الفترات الجزئية، باستخدام أسلوب الرصيد المتناقص المزدوج أو أسلوب آخر تحدده.

شكل الدالة :

VDB(cost, salvage, life, start_period, end_period, [factor], [no_switch])

Function Arguments

VDB

Cost	التكلفة	= number
Salvage	الخرده	= number
Life	العمر الانتاجي	= number
Start_period	بدء الاهلاك	= number
End_period	انتهاء الاهلاك	= number
Factor	المعامل	= number
No_switch	بدون تبديل	= logical

Cost is the initial cost of the asset.

Formula result =

[Help on this function](#)

OK Cancel

تتكون هذه الدالة من ٧ وسائط :

Cost تكلفة الاصل

Salvage القيمة عند نهاية الإهلاك (الخرده)

Life العمر الانتاجي

Start_period فترة بدء الاهلاك

End_period فترة نهاية الاهلاك

Factor : معامل التناقص ٢٠٠% او ١٥٠% وهكذا

No_switch : معامل اختياري

إذا كانت قيمة no_switch تساوي TRUE، او فارغة فانها تستخدم DDB او طريقة القسط المضاعف اما اذا كانت تساوي صفر 0 او FALSE فانها تستخدم SLN او القسط الثابت وليس المتناقص

المثال التالي يوضح طريقة عمل الدالة :

C	B	A	
	اثاثات المكتب	الأصل	1
	10000	تكلفة الأصل	2
	10	العمر بالسنوات	3
	1000	قيمة الخردة	4
			5
			6
	مجمع الاهلاك	قسط الاهلاك	السنة
2,000.00 ج.م.	2,000.00 ج.م.	1	8
3,600.00 ج.م.	1,600.00 ج.م.	2	9
4,880.00 ج.م.	1,280.00 ج.م.	3	10
5,904.00 ج.م.	1,024.00 ج.م.	4	11
6,723.20 ج.م.	819.20 ج.م.	5	12
7,378.56 ج.م.	655.36 ج.م.	6	13
7,902.85 ج.م.	524.29 ج.م.	7	14
8,322.28 ج.م.	419.43 ج.م.	8	15
8,661.14 ج.م.	338.86 ج.م.	9	16
9,000.00 ج.م.	338.86 ج.م.	10	17
			18
		لحساب قسط الاهلاك	19
	=VDB(\$B\$2,\$B\$4,\$B\$3,A8-1,A8)		20
		لحساب مجمع الاهلاك	21
	=VDB(\$B\$2,\$B\$4,\$B\$3,0,A8)		22

Function Arguments ? X

VDB

Cost	<input type="text" value="\$B\$2"/>	↑	= 10000
Salvage	<input type="text" value="\$B\$4"/>	↑	= 1000
Life	<input type="text" value="\$B\$3"/>	↑	= 10
Start_period	<input type="text" value="A8-1"/>	↑	= 0
End_period	<input type="text" value="A8"/>	↑	= 1

= 2000

Returns the depreciation of an asset for any period you specify, including partial periods, using the double-declining balance method or some other method you specify.

Cost is the initial cost of the asset.

Formula result = 2,000.00 ج.م.

[Help on this function](#) OK Cancel

ذكرنا في الدالة DDB ان في الحياه العملية يختلف معامل المضاعف فقد يكون ١٥٠% او ٢٠٠% كما في المثال التالي

اذا كان معامل التضاعف ١٥٠%

C	B	A	
	اثاثات المكتب	الأصل	1
	10000	تكلفة الأصل	2
	10	العمر بالسنوات	3
	1000	قيمة الخرده	4
	150%	المعامل	5
			6
	مجموع الاهلاك	قسط الاهلاك	السنة
	1,500.00 ج.م.	1,500.00 ج.م.	1
	2,775.00 ج.م.	1,275.00 ج.م.	2
	3,858.75 ج.م.	1,083.75 ج.م.	3
	4,779.94 ج.م.	921.19 ج.م.	4
	5,562.95 ج.م.	783.01 ج.م.	5
	6,250.36 ج.م.	687.41 ج.م.	6
	6,937.77 ج.م.	687.41 ج.م.	7
	7,625.18 ج.م.	687.41 ج.م.	8
	8,312.59 ج.م.	687.41 ج.م.	9
	9,000.00 ج.م.	687.41 ج.م.	10
	9,000.00 ج.م.		18
		لحساب قسط الاهلاك	19
	=VDB(\$B\$2,\$B\$4,\$B\$3,A8-1,A8,\$B\$5)		20
		لحساب مجموع الاهلاك	21
	=VDB(\$B\$2,\$B\$4,\$B\$3,0,A8,\$B\$5)		22

ملاحظة اذا ترك هذه المعامل فارغا فان الدالة تعتبر ان المضاعف ٢٠٠%

مثال اخر: اذا احتجت ان تحسب هذه الدالة بطريقة القسط الثابت عليك ان تضع مكان الوسيط no_switch القيمة FALSE

C	B	A	
	اثاثات المكتب	الأصل	1
	10000	تكلفة الأصل	2
	10	العمر بالسنوات	3
	1000	قيمة الخرده	4
	200%	المعامل	5
	FALSE	التبديل	6
	مجمع الاهلاك	قسط الاهلاك	7
900.00 ج.م.	900.00 ج.م.	1	8
1,800.00 ج.م.	900.00 ج.م.	2	9
2,700.00 ج.م.	900.00 ج.م.	3	10
3,600.00 ج.م.	900.00 ج.م.	4	11
4,500.00 ج.م.	900.00 ج.م.	5	12
5,400.00 ج.م.	900.00 ج.م.	6	13
6,300.00 ج.م.	900.00 ج.م.	7	14
7,200.00 ج.م.	900.00 ج.م.	8	15
8,100.00 ج.م.	900.00 ج.م.	9	16
9,000.00 ج.م.	900.00 ج.م.	10	17
	9,000.00 ج.م.		18
		لحساب قسط الاهلاك	19
		=VDB(\$B\$2,\$B\$4,\$B\$3,A8-1,A8,,\$B\$6)	20
		لحساب مجمع الاهلاك	21
		=VDB(\$B\$2,\$B\$4,\$B\$3,0,A8,,\$B\$6)	22

لاحظ هنا ان قسط الاهلاك اصبح ثابتا ومتساويا من فتره لآخرى


مثال اخر: في هذه الحالة المطلوب حساب الاهلاك بعد 3 سنوات لان الأصل سيبيع او يستبعد بعدها.

A	B	C	D	E
الأصل	اثاثات المكتب			
تكلفة الأصل	10000			
العمر بالسنوات	10			
قيمة الخرده	1000			
تاريخ الشراء	15/03/2013			
تاريخ البيع	15/03/2016	3	=DATEDIF(B5,B6,"y")	
المعامل	2			
بدون تبديل	TRUE			
قيمة البيع		5875		
التكلفة	10,000.00			
مجمع الاهلاك المتراكم	4,880.00			
صافي قيمة الأصل		5,120.00		
الربح عند بيع الأصل		755		

Source:Excel 2016 Formulas - Michael Alexander

مثال شامل مقارنة بين دوال الاهلاك SLN -DDB-DB-SYD-VDB

	A	B	C	D	E	F
1						
2	التكلفة	220,000				
3	الخرده	20,000				
4	العمر	4				
5						
6	أقساط الاهلاك					
7						
8	الفترة	SLN	SYD	DB	DDB	VDB
9	1	50,000	80,000	99,220	110,000	110,000
10	2	50,000	60,000	54,472	55,000	55,000
11	3	50,000	40,000	29,905	27,500	27,500
12	4	50,000	20,000	16,418	7,500	7,500
13						
14	القيمة الدفترية					
15						
16	الفترة	SLN	SYD	DB	DDB	VDB
17	0	220,000	220,000	220,000	220,000	220,000
18	1	170,000	140,000	120,780	110,000	110,000
19	2	120,000	80,000	66,308	55,000	55,000
20	3	70,000	40,000	36,403	27,500	27,500
21	4	20,000	20,000	19,985	20,000	20,000
22						



لاحظ ان الدالة SLN القسط الثابت يكون خطا مستقيما بينما باقي الدوال تعطي خطوطا منحنيه لان الاهلاك في السنوات الأولى يكون اكبر من السنوات التالية .

الدالة (1) AMORLINC للنظام المحاسبي الفرنسي

تُرجع الإهلاك لكل فترة محاسبية. يتم توفير هذه الدالة لنظام المحاسبة الفرنسي. وإذا تم شراء أحد الأصول في منتصف الفترة المحاسبية، فإنه يتم وضع الإهلاك المقسم بالتناسب في الاعتبار.

شكل الدالة :

AMORLINC(cost, date_purchased, first_period, salvage, period, rate, [basis])

Cost مطلوبة. وهي تكلفة الأصل.

Date_purchased مطلوبة. وهي تاريخ شراء الأصل.

First_period مطلوبة. وهي تاريخ نهاية الفترة الأولى.

Salvage مطلوبة. وهي القيمة الباقية في نهاية فترة عمر الأصل.

Period مطلوبة. وهي الفترة الزمنية.

Rate مطلوبة. وهي معدل الإهلاك.

Basis اختيارية. وهي أساس حساب السنة المستخدم.

أي تاريخ لابد أن يكتب بالرقم التسلسلي له أو كنتاج للدالة DATE(2008,5,23) والاسيقوم بإدخال التواريخ كنص لو أدخلت هكذا 1/1/2008.

▪ **Basis** اختيارية. وهي أساس حساب السنة المستخدم.

نظام التاريخ	Basis
360 يوم (أسلوب NASD)	0 أو مهمل
فعلي	1
365 يوم في السنة	3
360 يوم في السنة (الطريقة الأوروبية)	4

مثال :

	A	B	C	D	E	F	G
1	البيانات	الوصف					
2	2400	التكلفة					
3	١٩ أغسطس، ٢٠٠٨	تاريخ الشراء					
4	٣١ ديسمبر، ٢٠٠٨	نهاية الفترة الأولى					
5	300	قيمة الخردة					
6	1	الفترة الزمنية					
7	0.15	معدل الإهلاك					
8	1	الأساس الفعلي					
9							
10							
11	الاهلاك	360					
12							

الدالة (٢) AMORDEGRC للنظام المحاسبي الفرنسي

تُرجع الإهلاك لكل فترة محاسبية. يتم توفير هذه الدالة لنظام المحاسبة الفرنسي. وإذا تم شراء أحد الأصول في منتصف الفترة المحاسبية، فيتم وضع الإهلاك المقسم بالتناسب في الاعتبار. تتشابه هذه الدالة مع الدالة AMORLINC، باستثناء أنه يتم تطبيق **معامل الإهلاك** على الحساب استناداً إلى فترة عمر الأصول.

▪ معاملات الإهلاك هي:

معامل الإهلاك	فترة عمر الأصول (1/المعدل)
1,5	بين 3 و 4 سنوات
2	بين 5 و 6 سنوات
2,5	أكثر من 6 سنوات

	A	B	C	D	E	F	G
1	البيانات	الوصف					
2	2400	التكلفة					
3	١٩ أغسطس، ٢٠٠٨	تاريخ الشراء					
4	٣١ ديسمبر، ٢٠٠٨	نهاية الفترة الأولى					
5	300	قيمة الخردة					
6	1	الفترة الزمنية					
7	0.15	معدل الإهلاك					
8	1	الأساس الفعلي					
9							
10							
11	الاهلاك	776					
12							

امثلة عملية

على طريقة القسط الثابت والمتناقص المضاعف طبقا لمعيار المحاسبة الدولي رقم ١٦ والمعيار المصري رقم ١٠
(الأصول الثابتة واهلاكها)

مثال رقم ١

اشترت منشأة كرم جابر آلة بمبلغ ١٧٠٠٠ جنية وتبلغ القيمة المتبقية لها في نهاية عمرها الإنتاجي ٢٠٠٠ جنية والعمر المقدر لها ٥ سنوات.

والمطلوب:

١- حساب الاهلاك بطريقة القسط الثابت (SLN) وعمل جدول الاهلاك خلال ٥ سنوات

٢- حساب الاهلاك بطريقة القسط المتناقص في ظل افتراض ان

معدل التضاعف ٢ ومعدل التضاعف ١,٥

المطلوب الأول: حساب الاهلاك بطريقة القسط الثابت (SLN) وعمل جدول الاهلاك خلال ٥ سنوات

	A	B	C	D
1				
2	التكلفة	17000		
3	الخرده	2000		
4	العمر	5		
5				
6				
7	الفترة	مصروف الاهلاك	مجمع الاهلاك	صافي القيمة الدفترية
8	1	3,000.00 ج.م.	3,000.00 ج.م.	14,000.00 ج.م.
9	2	3,000.00 ج.م.	6,000.00 ج.م.	11,000.00 ج.م.
10	3	3,000.00 ج.م.	9,000.00 ج.م.	8,000.00 ج.م.
11	4	3,000.00 ج.م.	12,000.00 ج.م.	5,000.00 ج.م.
12	5	3,000.00 ج.م.	15,000.00 ج.م.	2,000.00 ج.م.
13				
14				
15	مصروف الاهلاك	=SLN(\$B\$2,\$B\$3,\$B\$4)		
16	مجمع الاهلاك	=SUM(\$B\$8:B8)		
17	صافي القيمة الدفترية	=B\$2-C8		

قسط الاهلاك الثابت = (التكلفة - الخرده) / العمر

المطلوب الثاني : حساب الاهلاك بطريقة القسط المتناقص في ظل افتراض ان معدل التضاعف ٢

B8		=DDB(\$B\$2,\$B\$3,\$B\$4,A8,2)		
A	B	C	D	E
1				
2	التكلفة	17000		
3	الخرده	2000		
4	العمر	5		
5				
6				
7	الفترة	مصرف الاهلاك	مجمع الاهلاك	صافي القيمة الدفترية
8	1	6,800.00 ج.م.	6,800.00 ج.م.	10,200.00 ج.م.
9	2	4,080.00 ج.م.	10,880.00 ج.م.	6,120.00 ج.م.
10	3	2,448.00 ج.م.	13,328.00 ج.م.	3,672.00 ج.م.
11	4	1,468.80 ج.م.	14,796.80 ج.م.	2,203.20 ج.م.
12	5	203.20 ج.م.	15,000.00 ج.م.	2,000.00 ج.م.
13				
14				
15	مصرف الاهلاك	=DDB(\$B\$2,\$B\$3,\$B\$4,A8,2)		
16	مجمع الاهلاك	=SUM(\$B\$8:B8)		
17	صافي القيمة الدفترية	=\$B\$2-C8		
18				

ومعدل التضاعف ١,٥

B8		=VDB(\$B\$2,\$B\$3,\$B\$4,A8-1,A8,1.5)			
A	B	C	D	E	F
1					
2	التكلفة	17000			
3	الخرده	2000			
4	العمر	5			
5					
6					
7	الفترة	مصرف الاهلاك	مجمع الاهلاك	صافي القيمة الدفترية	
8	1	5,100.00 ج.م.	5,100.00 ج.م.	11,900.00 ج.م.	
9	2	3,570.00 ج.م.	8,670.00 ج.م.	8,330.00 ج.م.	
10	3	2,499.00 ج.م.	11,169.00 ج.م.	5,831.00 ج.م.	
11	4	1,915.50 ج.م.	13,084.50 ج.م.	3,915.50 ج.م.	
12	5	1,915.50 ج.م.	15,000.00 ج.م.	2,000.00 ج.م.	
13					
14					
15	مصرف الاهلاك	=VDB(\$B\$2,\$B\$3,\$B\$4,A8-1,A8,1.5)			
16	مجمع الاهلاك	=SUM(\$B\$8:B8)			
17	صافي القيمة الدفترية	=\$B\$2-C8			
18					

مثال ٢ على الفترات الجزئية :

في ٢٠٠١/٦/١ اشترت شركة الياسمين الآلات ومعدات تكلفتها ١٦٠٠٠٠ جنية قيمة الخردة ١٠٠٠٠ والعمرا الافتراضي ٤ سنوات

بطريقة القسط الثابت وبطريقة القسط المتناقص المضاعف وبطريقة مجموع ارقام السنوات

أولا: بطريقة القسط الثابت :

	A	B	C	D	E	F	G	H	I	J
1	160000	تكلفة الأصل								
2	10000	قيمة الخردة								
3	4	العمر								
4	01/06/2001	البداية								
5	01/06/2005	النهاية								
6										
7		قسط الإهلاك								
8	1	21,875.00 ج.م.	21875							الفترة الأولى يدويا
9	2	37,500.00 ج.م.								
10	3	37,500.00 ج.م.	39750							بقاى الفترات
11	4	37,500.00 ج.م.								
12	5	15,625.00 ج.م.	16562.5							الفترة الاخيرها يدويا
13	Total	150000								
14	=IF(\$A8=1,SLN(\$A\$1,\$A\$2,\$A\$3)*((DATEDIF(\$A\$4,DATE(YEAR(\$A\$4),12,13),"m")+1)/12),									
15	IF(\$A8=\$A\$3+1,									
16	((12-(DATEDIF(\$A\$4,DATE(YEAR(\$A\$4),12,13),"m")+1))/12)*SLN(\$A\$1,\$A\$2,\$A\$3),SLN(\$A\$1,\$A\$2,\$A\$3)))									
17										
18										

المعادلة المستخدمة :

=IF(\$A8=1,SLN(\$A\$1,\$A\$2,\$A\$3)*((DATEDIF(\$A\$4,DATE(YEAR(\$A\$4),12,13),"m")+1)/12),
IF(\$A8=\$A\$3+1,
((12-
(DATEDIF(\$A\$4,DATE(YEAR(\$A\$4),12,13),"m")+1))/12)*SLN(\$A\$1,\$A\$2,\$A\$3),SLN(\$A\$1,\$A\$2,\$A\$3))
)

تم حساب الفترة الأولى :

=((160000-10000)/4)*7/12

تم حساب الفترة الأخيرة

=((160000-1000)/4)*5/12

ثانياً: بطريقة القسط المتناقص المضاعف

الفترة	قسط الإهلاك	طريقة حساب قسط الإهلاك لفترة جزئية	قسط الإهلاك لفترة جزئية	قسط الإهلاك لفترة كاملة
1	46667	=80000*7/12	46,666.67 ج.م.	80,000.00 ج.م. =DDB(\$A\$1,\$A\$2,\$A\$3,A8)
2	56667	=(80000*5/12)+(40000*7/12)	56,666.67 ج.م.	40,000.00 ج.م. =DDB(\$A\$1,\$A\$2,\$A\$3,A9)
3	28333	=(40000*5/12)+(20000*7/12)	28,333.33 ج.م.	20,000.00 ج.م. =DDB(\$A\$1,\$A\$2,\$A\$3,A10)
4	14167	=(20000*5/12)+(10000*7/12)	14,166.67 ج.م.	10,000.00 ج.م. =DDB(\$A\$1,\$A\$2,\$A\$3,A11)
5	4167	=10000*5/12	4,166.67 ج.م.	
Total	150000			

المعادلة المستخدمة :

=SUM(IF(A8=1,ROUND(DDB(\$A\$1,\$A\$2,\$A\$3,A8)*((DATEDIF(\$A\$4,DATE(YEAR(\$A\$4),12,31),"M")+1)/12),0),IFERROR(DDB(\$A\$1,\$A\$2,\$A\$3,\$A7)*(DATEDIF(DATE(YEAR(\$A\$4),1,1),DATE(YEAR(\$A\$4),MONTH(\$A\$4),DAY(\$A\$4)),"M")/12),0)),IF(A8>1,IFERROR(DDB(\$A\$1,\$A\$2,\$A\$3,\$A9-1)*((12-(DATEDIF(DATE(YEAR(\$A\$4),1,1),DATE(YEAR(\$A\$4),MONTH(\$A\$4),DAY(\$A\$4)),"M")))/12),0),0))

لو كانت الفتره كاملة :

قسط الإهلاك لفترة كاملة	السنة
80,000.00 ج.م. =DDB(\$A\$1,\$A\$2,\$A\$3,A8)	السنة ١
40,000.00 ج.م. =DDB(\$A\$1,\$A\$2,\$A\$3,A9)	السنة ٢
20,000.00 ج.م. =DDB(\$A\$1,\$A\$2,\$A\$3,A10)	السنة ٣
10,000.00 ج.م. =DDB(\$A\$1,\$A\$2,\$A\$3,A11)	السنة ٤

اهلاك الفترات الجزئية :

طريقة حساب قسط الإهلاك لفترة جزئية	قسط الإهلاك لفترة جزئية	فترة الإهلاك
=80000*7/12	46,666.67 ج.م.	1
=(80000*5/12)+(40000*7/12)	56,666.67 ج.م.	2
=(40000*5/12)+(20000*7/12)	28,333.33 ج.م.	3
=(20000*5/12)+(10000*7/12)	14,166.67 ج.م.	4
=10000*5/12	4,166.67 ج.م.	5
	150,000.00 ج.م.	المجموع

ثانياً: بطريقة مجموع ارقام السنوات

A	B	C	D	E	F	G	H	I
1	160000	تكلفة الأصل						
2	10000	قيمة الخردة						
3	4	العمر						
4	01/06/2001	البداية						
5	01/06/2005	النهاية						
6								
7		الفترة						
8	1	35000						
9	2	51250						
10	3	36250						
11	4	21250						
12	5	6250						
13	Total	150000						

$$=SUM(IF(A8=1,ROUND(SYD(A1,A2,A3,A8)*((DATEDIF(A4,DATE(YEAR(A4),12,31),"M")+1)/12),0),IFERROR(SYD(A1,A2,A3,$A7)*(DATEDIF(DATE(YEAR($A$4),1,1),DATE(YEAR($A$4),MONTH($A$4),DAY($A$4)),"M")/12),0)),IF(A8>1,IFERROR(SYD(A1,A2,A3,$A9-1)*((12-(DATEDIF(DATE(YEAR(A4),1,1),DATE(YEAR(A4),MONTH(A4),DAY(A4)),"M")))/12),0),0))$$

المعادلة المستخدمة

$$=SUM(IF(A8=1,ROUND(SYD(A1,A2,A3,A8)*((DATEDIF(A4,DATE(YEAR(A4),12,31),"M")+1)/12),0),IFERROR(SYD(A1,A2,A3,$A7)*(DATEDIF(DATE(YEAR($A$4),1,1),DATE(YEAR($A$4),MONTH($A$4),DAY($A$4)),"M")/12),0)),IF(A8>1,IFERROR(SYD(A1,A2,A3,$A9-1)*((12-(DATEDIF(DATE(YEAR(A4),1,1),DATE(YEAR(A4),MONTH(A4),DAY(A4)),"M")))/12),0),0))$$

لو كانت الفترة كاملة :

السنة	قسط الاهلاك لفترة كاملة
السنة ١	60,000.00 ج.م. =SYD(\$A\$1,\$A\$2,\$A\$3,A8)
السنة ٢	45,000.00 ج.م. =SYD(\$A\$1,\$A\$2,\$A\$3,A9)
السنة ٣	30,000.00 ج.م. =SYD(\$A\$1,\$A\$2,\$A\$3,A10)
السنة ٤	15,000.00 ج.م. =SYD(\$A\$1,\$A\$2,\$A\$3,A11)

لو كانت الفترة جزئية :

فترة الاهلاك	قسط الاهلاك لفترة جزئية	طريقة حساب قسط الاهلاك لفترة جزئية
1	35,000.00 ج.م.	$=60000*7/12$
2	51,250.00 ج.م.	$=(60000*5/12)+(45000*7/12)$
3	36,250.00 ج.م.	$=(45000*5/12)+(30000*7/12)$
4	21,250.00 ج.م.	$=(30000*5/12)+(15000*7/12)$
5	6,250.00 ج.م.	$=15000*5/12$
المجموع	150,000.00 ج.م.	

